Seven Things I've Learned About FM Synthesis

Dan Liszewski
BIG WINKY ENTERPRISES
bigwinky.com
May 24, 2016

Preface to May 2016 Revision

To learn about FM Synthesis, I have experimented with CSound's foscil opcode from time to time but, I never really developed a solid understanding. So I embarked on a small research project to better grasp the concept. This article is the result.

The present version corrects some errors and expands the narrative. From the beginning, the hope has been that sharing my exploration in to FM Synthesis would serve as a gentle introduction for others.

This paper presumes the reader is familiar with some mathematical concepts:

- Trigonometric functions: sine and cosine; what plots of these functions look like; how they relate to the unit circle
- Sine waves and their descriptions in terms of amplitude, frequency and phase
- The first year calculus topics: differentiation and integration of single variable functions

The author would someday like to create learning material that uses students' intuition, rather than formal math, to explain these ideas. Until then, time spent on Kahn Academy or with other internet resources may be helpful.

1) Equation for a Frequency Modulated Sine Wave

Frequency modulation synthesis, or *FM synthesis*, is about periodically varying the frequency of a tone, to produce additional tone colors (timbres). To keep things simple, our scope here only includes modulation of sine waves.

To develop an equation for FM sine wave, a simple sine wave tone will be our starting point. (See the excellent Wikipedia article https://en.wikipedia.org/wiki/Sine_wave) The simple sine wave has a fixed amplitude, frequency, and phase and, when used to drive audible vibrations of a listener's ear drum, there's no apparent tremolo or vibrato or complex tone-color sensed. (The Wikipedia page exposes an example pure sine wave tone that can be experienced using web multimedia.) An equation describing the simple sine wave is:

$$y(t)=A\sin(2\pi f\,t+\phi)$$

equation 1: Simple sine wave

where amplitude A, frequency f, and phase ϕ are constant.

Now, to apply frequency modulation to the signal in equation 1, we need to vary the frequency as time passes. Of course, there an infinitely many ways the frequency can be varied with time so we will introduce a function that specifies the instantaneous frequency. For our immediate purpose, we will further refine our instantaneous frequency function to be the sum of a fixed frequency and scaled time-varying function.

$$f(\tau)=f_c+f_\Delta x(\tau)$$

equation 2: frequency as a function of time

At this point, it is tempting to substitute this function directly in to our equation for a simple sine wave but, this would be an error. Before explaining why, an exploration of the concept of frequency itself is in order.

We correctly think of frequency as the number of times something happens per unit of time. To describe audio signals, frequency is often expressed as the number of times a repetitive signal repeats in one second. To describe a sine wave in terms of the unit circle, we think of frequency as how many rotations around the circle occur in a given amount of time. Maintaining that vision of the unit circle, we also have the position at any time given by an angle. Frequency then is the first derivative with respect to time of that angular position. (Compare this unit circle view to the linear motion scenario where linear velocity is the first derivative with respect to time of the linear position.) Since mathematical integration is the inverse of derivation we can know that, if we have velocity or frequency, we can use integration to find the position or angle.

Returning to the task of enhancing an equation for a simple sine-wave into one describing a FM sine-wave, we can now understand why we can't simply substitute our frequency function in to equation 1: the argument of the sine function always represents an angle so, we must take care to integrate our frequency function to obtain an angle quantity:

$$\omega=\frac{d\theta}{dt}=2\pi f$$

equation 3: frequency vs angle

$$\theta=\int_0^t 2\pi f(\tau)\,d\tau$$

equation 4: Angle as a function of time in terms of instantaneous frequency

This leads to our first goal, an equation describing frequency modulated sine-wave:

$$y_{FM}(t)=A\sin\left(\int_0^t 2\pi f(\tau)\,d\tau\right)$$

equation 5: Frequency Modulated Sine Wave

where $f(\tau)$ is the instantaneous frequency as a function of time. We'll find it helpful to describe instantaneous frequency as a sum of a fixed carrier frequency and a scaled modulating function:

$$f(\tau)=f_c+f_\Delta x(\tau)$$

equation 6: Modulation function as a sum of two components: a constant and a scaled function

where, f_c is the fixed carrier frequency, f_Δ is the deviation in frequency from the non modulated value and $x(\tau)$ is the modulating function that ranges from -1 to 1.

Note: y_{FM} degenerates to $y_{FM}=A\sin(2\pi f_c t)$ when there is no modulation and the instantaneous frequency is a constant equal to f_c .

The reader may be aware of FM radio transmitters and receivers. These devices are based on the same idea of modulating the frequency of a sine wave signal but, for the purpose of conveying information rather than generating sound. The communication technology uses a radio-frequency carrier as the fixed frequency and, an audio signal as the second component of the instantaneous frequency function. It is the job of the FM transmitter to embed the audio signal in to the carrier signal while the FM receiver separates the audio from the carrier. Note that since FM radio technology predates FM synthesis, the term carrier frequency is sometimes used to describe the fixed component of the instantaneous frequency function.

2) Equations for Simple FM

In the case of simple FM, instantaneous frequency will be described by sine wave, specifically we will write the function for instantaneous frequency as:

$$f(\tau)=f_c+f_\Delta\sin(2\pi f_m\tau)$$

equation 7: Frequency Modulated Sine Wave

Again, f_Δ is the deviation in frequency from the non modulated value. f_m is the modulation frequency.

Inserting this expression of instantaneous frequency in to our equation for the more general FM Sine wave gives us:

$$y_{FM}(t)=A\sin\left(\int_0^t 2\pi[f_c+f_\Delta\sin(2\pi f_m\tau)]\,d\tau\right)$$

equation 8: Simple FM Sine Wave

Evaluating the integral yields our second goal: an equation for simple FM sine-wave:

$$y_{FM}(t)=A\sin\left(2\pi f_c t-\frac{f_\Delta}{f_m}(\cos(2\pi f_m t))\right)$$

equation 9: Simple FM Sine Wave

3) Modulation Index

The modulation index indicates, as you might guess, the degree of modulation. For sine wave modulated frequency, the index is the ratio of the frequency deviation to the modulation frequency:

$$I=\frac{f_\Delta}{f_m}$$

equation 10: Modulation index: ratio of frequency deviation to modulation frequency

If we use this in the most recent equation above, we obtain:

$$y_{FM}(t)=A\sin(2\pi f_c t-I\cos(2\pi f_m t))$$

equation 11: Simple FM in terms of modulation index

4) Frequency Modulation and Phase Modulation are Essentially the same thing.

Well, that's an overstatement. They are not interchangeable. For example a radio receiver built to decode phase modulated signals, will not be able to decode a signal from an FM transmitter. But, for *synthesis* purposes we are interested in the *modulated* signal (y_{FM}), rather than the *modulating* signal ($x(\tau)$). In this case, we can achieve the same effects with PM that we can with FM, provided the modulating signal is sinusoidal.

In phase modulation (PM), we modulate the instantaneous phase of the signal. Given a modulating function $\phi(t)$ and a carrier of $A\sin(2\pi t)$, the modulated signal $y_{PM}(t)$ is given by

$$y_{PM}(t) = A\sin(2\pi f_c t + \phi(t))$$

equation 12: Phase modulation

Let us consider a sinusoidal modulating function $\phi(t) = \phi_\Delta \sin(2\pi f_m t)$ and use it in our expression for the phase modulated signal:

$$y_{PM}(t) = A\sin(2\pi f_c t + \phi_\Delta \sin(2\pi f_m t))$$

equation 13: Simple PM

Now consider this quote from the Wikipedia article on Phase modulation:

> *PM can thus be considered a special case of FM in which the carrier frequency modulation is given by the time derivative of the phase modulation*

For PM the phase at any point in time is given by the 'angle' part of y_{PM}

$$\phi_{PM} = 2\pi f_c t + \phi_\Delta \sin(2\pi f_m t)$$

equation 14: "angle" part of PM equation above

and we differentiate with respect to time to calculate the instantaneous frequency as:

$$f_{PM} = \frac{d\phi_{PM}}{dt} = 2\pi f_c + 2\pi f_m \phi_\Delta \cos(2\pi f_m t)$$

equation 15: frequency as derivative of angle

For FM, the phase at any point in time is the 'angle' part of y_{FM}

.

$$\phi_{FM} = 2\pi f_c t - I(\cos(2\pi f_m t) + 1)$$

equation 16: "angle" part of FM

and the instantaneous frequency:

$$f_{FM} = \frac{d\phi_{FM}}{dt} = 2\pi f_c + 2\pi f_m I \sin(2\pi f_m t)$$

equation 17: Simple FM frequency as derivative of phase

Let's look these these expressions for f_{PM} and f_{FM} together and remember that we are discussing FM and PM synthesis. In each method we are using a similar modulating function; but in FM we're modulating frequency and in PM we are modulating phase. We can see that ϕ_Δ plays a role in PM, that is similar to the role that modulation index, $I=\frac{f_\Delta}{f_m}$ plays in FM:

$$f_{PM}=\frac{d\phi_{PM}}{dt}=2\pi f_c+2\pi f_m \phi_\Delta \cos(2\pi f_m t)$$

equation 18: Simple FM frequency as derivative of phase

$$f_{FM}=\frac{d\phi_{FM}}{dt}=2\pi f_c+2\pi f_m I \sin(2\pi f_m t)$$

equation 19: Simple PM frequency as derivative of phase

5) Spectra of Frequency Modulated Signals

The effect of simple frequency modulation to the observed spectra, is the addition of sidebands at frequencies given by $f_c \pm n * f_m$.

There are theoretically an infinite number of sidebands. But almost all of the power of the resulting spectrum is contained in the first $I+1$ sidebands. This means when the modulation index is elevated, the power will be spread out to a larger number of sideband frequencies.

Care must be taken to insure that the frequency of the sidebands do not exceed the Nyquist criteria unless aliasing is a desired effect.

When the ratio of the modulation to carrier frequency is rational, sidebands belong to a harmonic series. If the ratio is irrational, the sound is metallic or possibly noisy at high levels of modulation index.

6) `foscil`

CSound's `foscil` opcode implements simple FM. The input parameters are:

`xamp`	amplitude
`kcps`	A common factor of both f_c and f_m
`kcarfac`	$f_c / kcps$
`kmodfac`	$f_m / kcps$
`kindex`	$I=f_\Delta / f_m$ (I is the modulation index)

`ifn`	function table number, should be a sine for simple FM
`iphase`	(optional)initial phase into lookup table

The perceived fundamental will be the minimum positive value of `kcps * (kcarfac – n * kmodfac)`, where `kcarfac` and `kmodfac` are integers and `n=1,2,3,...`

The `foscil` opcode is very handy for exploring simple FM with Csound. The simple instrument below demonstrates its use and can be used with the table above to test one's understanding of simple FM:

```
sr = 44100
kr = 4410
ksmps = 10
nchnls = 1

           instr 20
print   p5,   p6,  p7,     p8,   p9, p10
irise  = 0.020
ifall  = 0.020
iamp   = p4
icps   = p5
icarfac= p6
imodfac= p7
index  = p8
ifn    = p9 ; should be a sine wave

kenv   linseg  0,irise, 1, p3 - irise - ifall, 1,ifall, 0
ar     foscil  iamp, icps, icarfac,  imodfac, index, ifn, p10
       out      ar * kenv
           endin

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;for comparison, simple oscillator:

                      instr 10
irise  = 0.020
ifall  = 0.020
icps   = p5
ifn    = p6
iphs   = 0.618 ; golden ratio conjugate

kenv  linseg  0,irise, p4, p3 - irise - ifall, p4,ifall, 0

ar    oscil kenv,icps, ifn, 0


     out   ar
                      endin
```

The orchestra above is about as straight-forward as it can be. Here's a score that meanders around the state space of possible inputs to `foscil`:

```
f1 0 8192 10 1

i10 0   0.5  20000  256 1 ; middle C - just to orient the listener
s
f 0 1 ; dummy f statement - number of seconds to be silent.
s
;section 3
;p1 p2 p3     p4   p5   p6  p7     p8   p9 p10
;in st dur   amp   cps  car mod    indx fn phs
i20  0  0.1  20000  256  1   0      0     1  1 ; should sound like first tone
i20  +  .    .      256  .   .5     2     .  . ;increasing modfac
i20  +  .    .      256  .   <      .     .  .
i20  +  .    .      256  .   <      .     .  .
i20  +  .    .      256  .   <      .     .  .
i20  +  .    .      256  .   <      .     .  .
i20  +  .    .      256  .   <      .     .  .
i20  +  .    .      256  .   <      .     .  .
i20  +  .    .      256  .   <      .     .  .
i20  +  .    .      256  .   <      .     .  .
i20  +  0.5  .      256  .   17     .     .  .
s
f 0 1 ; dummy f statement - number of seconds to be silent.
s
;section 5
;p1 p2 p3     p4   p5   p6  p7     p8   p9 p10
;in st dur   amp   cps  car mod    indx fn phs
i20  0  0.1  20000  256  2   3      0     1  1 ;
i20  +  .    .      256  2   3      <     .  . ;
i20  +  .    .      256  .   .      <     .  . ;
i20  +  .    .      256  .   .      <     .  . ;
i20  +  .    .      256  .   .      <     .  . ;
i20  +  .    .      256  .   .      <     .  .
i20  +  .    .      256  .   .      <     .  .
i20  +  .    .      256  .   .      <     .  .
i20  +  .    .      256  .   .      <     .  .
i20  +  .    .      256  .   .      <     .  .
i20  +  .    .      256  .   .      <     .  .
i20  +  .    .      256  .   .      <     .  .
i20  +  1.4  .      256  .   .      40     .  .
s
f 0 1 ; dummy f statement - number of seconds to be silent.
s
;section 7
;p1 p2 p3     p4   p5   p6  p7     p8   p9 p10
;in st dur   amp   cps  car mod    indx fn phs
i20  0  0.1  20000  256  1   0      0     1  1 ; should sound like first tone
i20  +  .    .      256  2 0.3      5     .  . ; increasing carfac
i20  +  .    .      256  <   .      .     .  . ;
i20  +  .    .      256  <   .      .     .  . ;
i20  +  .    .      256  <   .      .     .  . ;
i20  +  .    .      256  <   .      .     .  . ;
i20  +  .    .      256  <   .      .     .  . ;
i20  +  .    .      256  <   .      .     .  . ;
i20  +  .    .      256  <   .      .     .  . ;
i20  +  .    .      256  <   .      .     .  . ;
i20  +  0.4  .      256  11  .      .     .  . ;
s
f 0 1 ; dummy f statement - number of seconds to be silent.
s
;section 9
;p1 p2 p3     p4   p5   p6  p7     p8   p9 p10
;in st dur   amp   cps  car mod    indx fn phs
i20  0  0.2  20000  256  1   0      0     1  1 ; should sound like first tone
i20  +  .    .      <    1.5 0.3    5     .  . ;
i20  +  .    .      <    .   .      .     .  . ; increasing cps
```

```
i20  +  .    .      <    .   .      .     .  . ;
i20  +  .    .      <    .   .      .     .  . ;
i20  +  .    .      <    .   .      .     .  . ;
i20  +  .    .      <    .   .      .     .  . ;
i20  +  .    .      <    .   .      .     .  . ;
i20  +  .    .      4000 .   .      .     .  . ;

s
f 0 2 ; dummy f statement - number of seconds to be silent.
s
;section 11
;p1 p2 p3     p4   p5   p6  p7     p8   p9 p10
;in st dur   amp   cps  car mod    indx fn phs
i20  0  0.15  20000  2720 1   0      0     1  1 ; different pitch
i20  +  .    .      .    . 0.3      5     .  .  ;
i20  +  .    .      .    .   .      50    .  . ;
i20  +  .    .      .    .   .      100   .  . ;
i20  +  .    .      .    .   6      5     .  . ;
i20  +  .    .      .    .   .      50    .  . ;
i20  +  .    .      .    .   .      100   .  . ;
i20  +  .    20000  100  1   0      0     1  1 ; different pitch
i20  +  .    .      .    . 0.3      5     .  .  ;
i20  +  .    .      .    .   .      50    .  . ;
i20  +  .    .      .    .   .      100   .  . ;
i20  +  .    .      .    .   6      5     .  . ;
i20  +  .    .      .    .   .      50    .  . ;
i20  +  .    .      .    .   .      100   .  . ;

s
f 0 1 ; dummy f statement - number of seconds to be silent.
s
;section 13
;p1 p2 p3     p4   p5   p6  p7     p8   p9 p10
;in st dur   amp   cps  car mod    indx fn phs
i20  0  1.8  20000  256 31   11      3     1  1 ; according to forumla: this should have
same perceived fundamental as next note:

i10 2   1.8  20000  2304 1 ;

s
f 0 2 ; dummy f statement - number of seconds to be silent.
s
;section 15
;p1 p2 p3     p4   p5   p6  p7     p8       p9 p10
;in st dur   amp   cps  car mod    indx     fn phs
i20  0  0.15 20000  60   1   4      0.3     1  1
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
i20  +  .    .      <    <   <      <       .  .  ;
```

```
i20  +  .    .     <    <   <      <       .  .  ;
i20  +  .    .     3333 11  1.1    6       .  .  ;

s

e
```

This score consists of 15 sections. The even numbered sections are silence used to separate the (non-silent) experiments contained in the odd numbered sections. If you real-time render this score using the orchestra above, the section numbers appear in the console as they are played.

Section	**Intention**	**Observations**
1	A 256Hz (middle C) pure sine – for comparison to subsequent FM tones	The classic "Hearing test" tone
3	Series of tones with increasing `modfac` – while keeping ever other parameter constant.	Side bands are increasing in frequency. 256 Hz sine is very audible.
5	Series of tones with increasing index.	Becomes more "buzzy" with each increase. The warmth of the original 256Hz sine is only audible in the first tone. On frequency plot, more side bands are present with higher index.
7	Series of tones with increasing `carfac`	Fundamental and side bands are shifted higher with each tone.
9	Series of tones with increasing cps	Tones seem un-related, don't "echo" like other examples.
11	Arbitrary modification of cps, `modfac` and index.	These 14 tones make an interesting loop! Bells, warbles, and buzzing are all heard.
13	A test of the statement : The perceived fundamental will be the minimum positive value of `kcps * (kcarfac – n * kmodfac),` where `kcarfac` and `kmodfac` are integers and `n=1,2,3,...` The first tone is FM, the second, a pure sinusoid at a frequency given by the formula.	Almost! The overtones are so strong, it's hard to isolate a single "perceived fundamental"
15	Arbitrary modification of cps, `modfac` and index.	These 26 tones make an interesting loop! Bells, whistles, buzzing, and some lower frequency warm tones are all heard.

7) What to Learn Next

The basic FM idea can be extended by:

- Modulating the modulation function (cascade) to create additional sidebands
- Feedback (self modulation)
- Summing multiple sinusoidal modulators (Parallel modulators)
- Non sinusoidal carriers and/or modulators: noise, audio, etc...
- Interactive FM explorer implemented in PureData

While some of these algorithms are available in commercial synthesizers, all of them provide the FM synthesis student inspiration for further study and experimentation.

References

Benson, Dave. Music: a Mathematical Offering. Cambridge: Cambridge UP, 2007. Music: a Mathematical Offering. 9 Mar. 2008 <http://www.maths.abdn.ac.uk/~bensondj/html/music.pdf>.

Lesurf, Jim. "Frequency Modulation, Phase Modulation, and FM Spectra." Scot's Guide to Electronics. 15 June 2005. University of St. Andrews. 5 Mar. 2008 <http://www.st-andrews.ac.uk/~jcgl/Scots_Guide/RadCom/part12/page1.html>.

Pinkston, Russell. "FM Synthesis in CSound." The CSound Book. Ed. Richard Boulanger. Cambridge, Massachusetts: The MIT P, 2000. 261-270.

Puckette, Miller S. The Theory and Technique of Electronic Music. World Scientific, 2007. The Theory and Technique of Electronic Music. Oct. 2006 <http://crca.ucsd.edu/~msp/techniques/v0.09/book.pdf>.

Reid, Gordon. "Synth Secrets." Sound on Sound. Apr. 2000. 9 Mar. 2008 <http://www.soundonsound.com/sos/apr00/articles/synthsecrets.htm>.

Schottstaedt, Bill. "An Introduction to FM." CCRMA. 11 Feb. 2008. Stanford University. 7 Mar. 2008 <http://ccrma.stanford.edu/software/snd/snd/fm.html>.

Vercoe, Barry, and Et Al. "Foscil." Csounds.Com. 1992. MIT Media Lab. 10 Mar. 2008 <http://www.csounds.com/manual/html/foscil.html>.

Wikipedia contributors. “Phase modulation” [Internet]. *Wikipedia, The Free Encyclopedia*; 08:10 16 February 2008 UTC<http://en.wikipedia.org/wiki/Phase_modulation> [accessed 2008 Mar 7].

Wikipedia contributors. “Frequency modulation”, *Wikipedia, The Free Encyclopedia*, 21:37 2 March 2008 UTC <http://en.wikipedia.org/wiki/Frequency_modulation> [accessed 2008 Mar 7].